PROC GCHARTS 

Bar Charts

Bar charts are used frequently.  You have been introduced to bar charts already.  There are some special features that can be used with PROC GCHARTS.  The main features are PATTERN and VALUE statements.  You will also see many other features that you can use.  

When you make a graph, SAS will fill in the bars with default designs--usually cross hatching.  SAS may also choose colors if none are listed. If you want some control over the patterns and colors in the bars, you can use the PATTERN statement.  It consists of the PATTERN statement and the VALUE statement. 

Each of these graphs will use the same font for titles and for all other text on the charts.  It is important to have good readable text that is large and clear and uncluttered.  You can define them  at the beginning of your program by using a GOPTIONS statement.  The GOPTIONS is a global statement and will be used for all graphs unless you change it in the program.  All of the programs used in these examples use:

Goptions ftitle = swiss ftext = zapf; 

Goptions border;

Goptions rotate = landscape;
The ftitle  specifies what the font will be for the title in each graph.  The ftext  specifies what all other text will be in the graph.  This means the axis text, labels, and legends will use this second font.  The font, swiss, is a clear font and is dark and will stand out in the titles. The font, zapf,  is a darker font and shows up better on graphs.  This is helpful when you are making  overheads as the print stands out more.  

The border option is used to put a border around the entire graph to make it look better for the graphs in this book.   The graphs will also be printed out in a landscape form. 

All data used for these graphs will be at the end of the unit on SAS/GRAPHS.  Examples will start with a simple graph.  More features will be added as you go along. Here is the first example:

data all;

infile 'a:coll2002.dat';

input  year 1-4 college $ 6-7 total 9-12;

label total = 'Total Students'

         college = ‘College’;

if year = 2002;

pattern value=solid c=red;
proc gchart ;
  
 vbar college / sumvar = total noframe;
title c=red 'OSU College Enrollments';

title2   c=black 'Fall 2002';

footnote c=black j=left 'Source: OSU Student Profile, Fall, 2002';

footnote2 f=swissb 'Output 2.1';

run;

Notice that after the 'vbar college', the 'sumvar=total' was used.  In this particular case, only the summary data was available--not an entire data set. The sumvar statement had to be used here.  All of these graphs used summary data.  They would all contain the sumvar=total.  The sumvar=total is only used in this first chart so you can see how to work with summary data.    If you had a regular large data set, you would not use the sumvar statement. The line would read "vbar college;”   All other examples will assume that you have an entire data set. 

The noframe was used in this example because SAS puts a default frame around the graph itself (where the bars are shown) if you don’t specify the “noframe”.  You will see the frame in another example.  

Pattern statements should always go before the proc gchart statement.  Since there is only one Pattern statement here, all of the bars will have the same pattern.  The "value=solid" means that the bar will be completely filled in.  You can use v = for short. 

The values that are acceptable are:
1:  X1 - X5


4:  S

2:  L1 - L5


5:  E

3:  R1 - R5

The X stands for cross hatching.  The 1 to 5 explain how thick you want the fill.  The "1" is the thinnest and the "5" is the thickest.

The L stands for parallel lines that go to the LEFT.  The R stands for parallel lines that go to the RIGHT.

The S means that the bar will be filled solid.  The E means that the bar will remain empty.

You can use several pattern statements and also add a color statement.  This next example uses different patterns and also subgroups the bars by class.  

OUTPUT 2.1     SIMPLE VBAR CHART

Some other features in the next program are the use of the comma in the total number of students.  The proc format statement is used to change the class variable from numbers to word descriptions.  A box is also used around the legend -- the definition of class in this case.


data two;

infile 'a:class2002.dat’;

input college $ 1-2 class 5 total 7-10;

label total = “Total Students”

          class = “Class” 

         college = “College”;
proc format;

   

value numbers

 


1 = 'Freshmen'

       


2 = 'Sophomores'

       


3 = 'Juniors'

       


4 = 'Seniors';
pattern1 v=x3 c=red;

pattern2 v=r4 c=blue;

pattern3 v=x5 c=green;

pattern4 v=l5 c=magenta;
legend frame label = none;
proc gchart;

 

 vbar college / subgroup=class sumvar=total noframe legend = legend;
format class numbers.  total comma7.;
title c=red 'OSU Enrollments by Class for Fall, 2002';

footnote j=l   'Source: OSU Student Profile, Fall, 2002';

footnote2 f=swissb 'Output 2.2';

run;

OUTPUT 2.2   SUBGROUPED CHART

There is also a 3D feature that will allow you to make a 3d chart.  This next example uses the same information as in the example above except for the vbar3d statement. Here is the program:

pattern1 v=x3 c=red;

pattern2 v=r3 c=blue;

pattern3 v=x5 c=green;

pattern4 v=l5 c=magenta;

legend frame label = none;

proc gchart;

 

 vbar3d college / subgroup=class sumvar=total noframe legend =legend;
format class numbers.  total comma7.;

title c=red 'OSU Enrollments by Class for Fall, 2002';

footnote j=l   'Source: OSU Student Profile, Fall, 2002';

footnote2 f=swissb 'Output 2.3';

run;


Note that the 3D chart did not use the patterns that were provided.  It did use the colors that were listed.  Apparently, the 3D chart uses the solid pattern on charts.  This 3D feature is fairly new and I have not seen any information on it in any of the SAS/GRAPH books I have.

OUTPUT 2.3   3D CHART

Part of learning to make graphs is to also learn about extra features that can be added to graphs to incorporate extra information.  This next example adds a few new features to the graph that was developed above.  These are the use of a frame around the graph itself.  The other is the use of the sum statement where the numbers appear above each of the bars. The bars are plotted in descending order.  The comma feature is also being used.  If you had many bars, you could not use the comma feature as the graph would be too crowded.     Look at the differences in these graphs

pattern1 v=x3 c=red;


pattern2 v=r3 c=blue;

pattern3 v=x5 c=green;

pattern4 v=l5 c=magenta;

proc gchart;

 

 vbar college/subgroup=class sumvar=total sum descending;

format class numbers. total comma6. ;   
title f=swiss c=red 'OSU Enrollments by Class for Fall, 2002';

footnote j=l f=swiss ‘Source: OSU Student Profile, Fall 2002;

footnote2 f=swissb 'Output 2.4';

run;

See OUTPUT 2.4a.

The noframe option was deleted from this graph.  If you do not put it in the program, it will automatically put a frame around the bars.   Notice that there is a frame around this graph. 

NOTE:  If you try out the graph above and the numbers do not appear above the bars, it is because the bars are not wide enough.  You can fix this by adding the WIDTH feature.  This allows you to set the width of the bars.  You can change it to different widths. You may need to experiment with different widths to get the results you want.   Here is an example of the vbar statement with the WIDTH feature:

vbar college / subgroup=class sumvar=total sum descending width = 7;
See OUTPUT 2.4b.

OUTPUT 2.4a

OUTPUT 2.4b

This next bar chart uses the group feature.  The proc format is used here to change the short names of the colleges to more meaningful names.  NOTE:  When you have a proc format statement that is character (in this case--$univ), you must use the $ sign when you are describing it in the format statement.  This particular program also uses the subgroup statement to change the colors of each year.  The original data set also had information on other colleges besides the three listed here.  The where statement is used to determine which data from the file is to be plotted.
data major;

infile 'a:majors2002.dat';

input year 1‑4 college $6‑7 total 9‑12;

proc format;

  

 value $univ

      

'AS' = 'Arts & Sciences'


'BU' = 'Business'

      

'EN' = 'Engineering';
label total = 'Total';

pattern1 v=r3 c=red;

pattern2 v=x2 c=blue;

pattern3 v=s c=green;

pattern4 v=l4 c=cyan;

pattern5 v=x4 c=black;

pattern6 v=s c=magenta;

proc gchart;
   

where college = 'AS' or college = 'BU' or college = 'EN';
    

 vbar year / discrete sumvar=total group=college subgroup=year;

format college $univ.;
title 'Oklahoma State University';

title2 'Comparison of Undergraduate Enrollments for 1997 to 2002';

footnote ‘Source: OSU Student Profile, Fall 2002’;

footnote2 f=swissb 'Output 2.5';

run;

OUTPUT 2.5

Next is a horizontal bar chart.  It has the same information as the one above only the bars are now horizontal.  Notice that the totals are out to the side of each bar.  The use of legend in a box is also used here.  One problem that you will encounter with several bars of different designs is that you can create an optical illusion that the bars are crooked.  This is particularly a problem with a small graph.  If this was in color, you could make more of the bars a solid color and avoid this problem.  The hbar follows.
data major2;

set major;

proc format;

   

value $univ

     

'AS' = 'Arts & Sciences'



'BU' = 'Business'

       

'EN' = 'Engineering';

pattern1 v=r3 c=red;

pattern2 v=x2 c=blue;

pattern3 v=s c=green;

pattern4 v=x4 c=cyan;

pattern5 v=r4 c=black;

pattern6 v=s c=magenta;

legend frame  label =none;
proc gchart;

   

where college = 'AS' or college = 'BU' or college = 'EN';

    
hbar year / discrete group=college subgroup=year sumvar=total legend=legend;
format total comma7. college $univ.;
title 'Oklahoma State University';

title2 'Comparison of Undergraduate Enrollments for 1997 to 2002';

footnote2 f=swissb 'Output 2.6';   

run;

OUTPUT 2.6

NOTE:  Occasionally when you make a horizontal bar chart there does not seem to be much space between the groups of bars and it is difficult to see the breaks in the groups.  The graph above was grouped by college.   You can add more space between the groups by using the GSPACE feature.   This may happen more with the horizontal bar charts.  An example of its use follows. 

hbar year /discrete group=college subgroup=year gspace = 2 sumvar=total legend=legend;

There are a few other features that you may want to try.  When we first discussed charts, you had an example of data that had a column called “Size”.  This was character data.   Three different bars were described, Short, Average, and Tall.  When SAS displays them, it puts them in alphabetical order—Average, Short, and Tall.   If you want it in another order, you can use the MIDPOINTS= option.  You can also change the colors on the graph using a PATTERNID option. This can be used with midpoints and also group.  Each group would be a different color.  Here is an example of a program that will use both of these features.

Data wtloss;

Set healthy;


Length size $7.;

If height <= 64 then size = ‘SHORT’;


If height >= 65 and height <= 70 then size = ‘AVERAGE’;


If height > 70 then size = ‘TALL’;


Label size = ‘SIZE’;



Pattern v=s c=red;


Pattern2 v=s c=green;


Pattern3 v=s c=blue;


Proc gchart;

Vbar size / midpoints = ‘SHORT’ ‘AVERAGE’ ‘TALL’ patternid=midpoint noframe width=10; 


Title ‘USE OF MIDPOINTS TO CONTROL ORDER OF CHARACTER DATA’;


Title2 ‘USE OF PATTERNID TO CHANGE COLORS OF EACH BAR’;


Footnote f=swissb ‘Output 2.7’;

OUTPUT 2.7


The next one uses two colors for a group.  See the following program:

Proc gchart;

Vbar size / midpoints = ‘SHORT’ ‘AVERAGE’ ‘TALL’ group = sex                                       patternid=group noframe width=8; 


Title ‘GRAPH OF SIZE GROUPED BY SEX’;


Title2 ‘USE OF PATTERNID WITH GROUP’;


Footnote f=swissb ‘Output 2.8’;


Run;
[image: image16.jpg]Degrees Granted at OSU for 2002—2003

Undergraduate and Graduate

degree=BACHELORS degree=GRADUATE

as
ars a6
186

BU
353

Source: OSU Student Profle, Fall 2002
Output 212

EN
389




OUTPUT 2.8

Pie Charts 

Pie charts only use one column.  They emphasize the whole of that particular column.  Pie charts have some more options. They also use various patterns.  These take into consideration different angles based on 360 degrees.  The VALUE pattern can use the following:

1.  PEMPTY   or PE

2.  PSOLID    or PS  

3.  P density<style<angle>>

where 

density can be 1 ...5 

(The thickness of the lines)

style can be X or N

(X is cross hatching, N is parallel lines)

angle can be 0 ... 360

(the angles can be in degrees)

Some other features of the PIE are being able to change the location of the name of the variables and the amount of the section of the Pie.   

type= The default is to give freq statistics.  You can use type=percent to obtain the percentages.  
Some other features are the slice and value, fill, explode, and invisible.  
slice can be used to place the slice labels inside the slices.  Use slice=inside.  If it is omitted, the slice labels go outside.

value can be used with arrow.  This creates an arrow to connect the amounts with their corresponding slices.  


fill allows you to fill in the slices with Solid.  The fill=solid will fill in each slice with default colors from SAS.  You can also use fill=x for cross hatching.

explode can be used to offset one particular slice.

invisible allows you to leave one of the slices empty.

across allows you to put two or three pies across a page (landscape format)

down allows you to put two or three pies down the page (portrait format)

This first program uses different pattern statements and the value=arrow.  The names of the colleges are changed to spell out their longer names.  Unfortunately, these truncate at 16 characters.  

data degree;

infile 'a:degree2002.dat';

input college $ 1‑2 degree $ 4‑5 total 6‑9;

proc format;

   

value $name

      


'AG' = 'Agriculture Sci'

     

 
'AS' = 'Arts & Sciences'

      


'BU' = 'Business Admin'

      


'ED' = 'Education'

      


'EN' = 'Engineering'

      


'HE' = 'Human Envir Sci';

pattern1 v=p4n90 c=red;

pattern2 v=ps c=cyan;


pattern3 v=p3n90 c=black;

pattern4 v=ps c=magenta;

pattern5 v=p4x0 c=blue;

pattern6 v=ps c=yellow;

proc gchart;

   

where degree = 'BA';

 

pie college / sumvar =total 





value = arrow noheading;
format college $name.;

title 'Degrees Granted at OSU for 2002‑2003';

title2 'Undergraduate Students';

footnote j=l ‘Source: OSU Student Profile, Fall 2002’;

footnote2 f=swissb 'Output 2.9'; 
run;

OUTPUT 2.9

The pie chart also will allow you to make a 3D graph.  The same graph that was used above will be used for the 3D graph.  Notice that even though there were patterns used, all of the pieces of the pie are filled in with solid colors.  One other feature is being used here.  It is MATCHCOLOR.  It will use the same color to print out the name and number of each piece of pie.
pattern1 v=p4n90 c=red;

pattern2 v=ps c=cyan;


pattern3 v=p3n90 c=black;

pattern4 v=ps c=magenta;

pattern5 v=p4x0 c=blue;

pattern6 v=ps c=green;

proc gchart;



where degree = 'BA';



pie3D college / sumvar =total matchcolor value = arrow noheading;
format college $name.;

title 'Degrees Granted at OSU for 2002-2003';

title2 'Undergraduate Students';

footnote j = l ‘Source: OSU Student Profile, Fall 2002’;

footnote2 f=swissb 'Output 2.10';

run;

OUTPUT 2.10

You can also display two or more pie charts on the same page.  You use the group= and across=2 command.  SAS also puts in its own heading on the graphs.  If you want that removed, you can use noheading.      A program follows. 

data degree2;

set degree;

proc format;



value $grad




'BA' = 'BACHELORS'




'CO' = 'GRADUATE';

pattern1 v=p4n90 c=red;

pattern2 v=ps c=cyan;

pattern3 v=p3n90 c=black;

pattern4 v=ps c=magenta;

pattern5 v=p4x0 c=blue;

pattern6 v=ps c=green;

proc gchart;


pie college / sumvar = total explode = 'AS'




slice= outside




group= degree




across = 2




noheading;
format degree $grad.;

title c=blue 'Degrees Granted at OSU for 2002-2003’;

title2 c=green 'Undergraduate and Graduate';

footnote j=l f=swiss ‘Source: OSU Student Profile, Fall 2002';

footnote2 f=swissb 'Output 2.11’;

run; 

OUTPUT 2.11

Notice that there is a problem when the “AS” piece of pie is exploded.  The names of the pieces of the pie (AS) cut into the titles of the two pies (Bachelors and Graduate).  One way to fix this would be to delete the “explode” on the pie.  Another way is to change the angle of the pie.   

By default, the first slice in a pie chart starts at the 3 o’clock position.  Then the other pieces go counterclockwise.  You can change the starting position with the ANGLE option.  You can start the first slice of a pie at the top (12 o’clock) by using ANGLE=90. The pie can use several angles – 0 to 360.   Pieces of the pie are always listed in either alphabetical order or numeric order counterclockwise.  

Changing the angle of the pie in the last graph will allow you to fit the slices on the graph without the name running into the label title.  See the following graph for an example. 

proc gchart;


pie college/ sumvar=total explode = 'AS'




slice= outside




group= degree




across = 2




angle = 45
noheading;

format degree $grad.;

title c=blue 'Degrees Granted at OSU for 2001-2002';

title2 c=green 'Undergraduate and Graduate';

footnote j=l f=swiss ‘Source: OSU Student Profile, Fall 2002’;

footnote2 f=swissb 'Output 2.12’;

run;

OUTPUT 2.12


Instead of “across =”, you can use “down = “.   The graphs will have two graphs one underneath the other one.  This works when you are using portrait to print out your graphs. 

You can use both the “across=” and the “down =” on the same graph. You would have four graphs on the same page.  Notice that when there are two graphs on a page, you cannot use the label names for the columns.  If you tried to use the label names for the columns, they would take up all the space and the pie itself would be very small!!

Using gray shades for the graphs instead of color


When you give a presentation to a group of people, you will want to use graphs that are done in color.  However, when you have them in a publication, you cannot always use color since it would be too expensive to duplicate.  In your class notes for SAS, I make the graphs in color.  But they are copied in black and white because it would be more expensive if I had the graphs duplicated in color for your notes.  If you have to publish something, you need to use colors that will copy well.  One way to do this is to use gray colors.   SAS has several colors that can be used.   You may have to try out several colors to see what will work.   The pie chart is one problem if you use all solid colors in the pieces.  They may blend together when they are copied and you can’t tell where one piece ends and another begins.  Using patterns between solid slices of the pie can help.

Some of the gray patterns are listed as:



Grayaa    /* This is the darkest gray   */



Graybb



Graycc



Graydd



Grayee



Grayff 
 /* This is white    */

The colors in the middle get lighter as they go from aa to ff.  There is one new feature that has been added here.  It is the COUTLINE.  This will put a colored outline around each bar and it will be colored in black.  The reason for the use of the coutline is that the last color, grayff, will be white and you would not see any bar if you did not use the coutline.  The following is a bar chart that uses all of these.  

pattern v=s c=grayaa; /* dark gray */

pattern2 v=s c=graybb;    

pattern3 v=s c=graycc;

pattern4 v=s c=graydd;

pattern5 v=s c=grayee;

pattern6 v=s c=grayff;    /* white */

proc gchart;

   vbar college/subgroup=college sumvar = total 

             width=6 noframe coutline=black;
title c=red 'OSU College Enrollments';

title2   c=black 'Fall 2002’;

footnote j=l c=black ‘Source: OSU Student Profile, Fall, 2002’;

footnote2 f=swissb 'Output 2.13';

run;


The next one is a pie chart.  It will use gray colors that have numbers attached to them instead of alphabetic characters.  They range from 22 to 99. Try to put a dark one next to a light one so that there is contrast. 

pattern1 v=s c=gray22;   /*  Darkest gray   */

pattern2 v=s c=gray88;   

pattern3 v=s c=gray33;

pattern4 v=s c=gray99;   /* Lightest gray   */

pattern5 v=s c=gray44;

pattern6 v=s c=gray77;  

proc gchart;

   where degree = 'BA';

 pie college / sumvar = total

    value = arrow noheader;

format college $name.;

title 'Degrees Granted at OSU for 2002-2003;

title2 'Undergraduate Students';

footnote j=l f=swiss ‘Source: OSU Student Profile, Fall 2002;

footnote2 f=swissb 'Output 2.14';

run;

OUTPUT 2.13

Review of Features Used in Proc Gchart

These are some features used in PROC GCHART:


Proc gchart;



Vbar column / options;      (also vbar3d)  



Hbar column / options;



Pie column /options;           (also pie3d)

Options for Vbars and Hbars; 

Subgroup =         
- Stacked bars with different colors 


Group = 
    
- Side by side charts 


Discrete 

- Used with numeric data to eliminate continuous numbers


Pattern value =   
- Vary from L1-L5, R1-R5, X1-X5, S, E 





(Parallel lines to Cross hatching, to Solid to Empty)


Color = (or c=)  
 - color 


Sumvar =           
 - Used for summary data when you don’t have an entire data set

Noframe             
- Used when you don’t want a frame around the graph –                 default is frame

Sum                   
- Adds the sum above a bar in a bar chart

Width =             
-  Width makes bars wider

Gspace =         
- Allows more space between groups in charts

Descending        
-  Bar charts go in descending order –Highest to Lowest

Ascending         
-  Bar charts go in ascending order – Lowest to Highest

Legend frame

- Puts a frame around the legend in the graph

Midpoints = 

- Used to change order of character data in a graph 

Patternid = 

- Used to give different patterns to each bar 

Type = 

- Used for percent, frequency, mean

Options for Pie charts:



Pattern value= 


- Vary from 




PE (Empty)




PS (Solid)




Pd (density –thickness of line)

Pds (style can be X-Cross-hatching or N-Parallel lines)


Pdsa (angle can be 0-360 degrees)


(example P5X45)

slice =


- Place slice labels inside pie or outside

value = arrow

- connects the amounts with the slices

fill = 


- fill=solid - fills in each pie slice with a SAS default color 

explode =

- Offsets one of the slices of pie

Invisible =

- Leaves one of the slices empty

Across = 

- Allows you to put two or three pies across a page

Down = 

- Allows you to put to or three pies down a page

Group = 

- Used to group pies across or down a page

Noheading 

- Leaves off the default heading on pie charts

Matchcolor

- The color of the print matches the color of the slice of pie

Angle =

- Change the angle where the pie starts (normally at
   3 o’clock position)

